
Abstract. The Viterbi algorithm is a classical example of a dynamic pro-
gramming algorithm, in which pruning reduces the search space drastically,
so that an otherwise exponential time complexity is reduced to linearity. The
central steps of the algorithm, expansion and pruning, can be expressed in
a concise and clear way in CHR, but additional control is needed in order
to obtain the desired time complexity. It is shown how auxiliary constraints,
called trigger constraints, can be applied to fine-tune the order of CHR rule
applications in order to reach this goal. It is indicated how properties such
as confluence can be useful for showing such optimized programs correct.

1 Introduction

Hidden Markov Models (HMMs) are probabilistic finite state machines that for
each transition emits a symbol from a finite alphabet, also by probabilistic choice.
HMMs are commonly used for modeling and analysis of, e.g., biological sequence
data and for speech recognition; see, e.g., [8,12]. Given a specific HMM and an
observed sequence over the alphabet, prediction means to find the most probable
path, i.e., sequences of states, by means of which the sequence may have been
produced; such a path is called a Viterbi path. Informally speaking, a Viterbi path
represents the most feasible interpretation or explanation of the given sequence; in
the biological case, the sequence may be DNA and the Viterbi path indicates the
most believable shifts between coding and non-coding regions, and perhaps details
concerning introns and exons [8].

HMMs owe much of their popularity to the existence of efficient algorithms for
training and, as we consider here, prediction in terms of the classical Viterbi algo-
rithm [17]. It is a dynamic programming algorithm that gradually extends optimal
paths so they cover a longer and longer prefix of the sequence, and eventually the
entire sequence. The algorithm keeps track of one optimal path ending in each state
s for the sub-sequence seen so far; call this set of partial paths Σ. Any non-optimal
path is discarded. In the next step, a new set of optimal paths is found among
the possible extensions of any σ ∈ Σ with one more state. The time complexity is
O(n · k2) and the space complexity is O(n · k) where n is the sequence length and
k the number of states.

In this paper, we investigate how well the Viterbi algorithm can be expressed
in CHR, considering both conciseness and efficiency. CHR, or Constraint Handling
Rules [9,10], was introduced as a declarative language for writing constraint solvers,
but has shown to be very useful for a variety of automated reasoning tasks, and
attempts have been made to use it as a general language for describing algorithms.

We show that the fundamental steps in the Viterbi algorithm can be exposed
very clearly in CHR, but to reach the optimal time complexity, we need to intro-
duce some techniques. We suggest to use trigger constraints, by means of which
a program’s operational behavior can be fine-tuned. For confluent programs, this
can be analyzed in a systematic way, and as a more general case, we put forward
informally the notion of “relative confluence” based on a more flexible state equiv-
alence (as compared with the usual logical equivalence of states). However, in order
to reach the optimal time complexity, we need to make additional transformations
that reflect the underlying CHR system’s search and matching. This may be less
satisfactory from the point of view of declarative programming, but may inspire to
the development of new automatic analyses and transformations to be included in
CHR implementations.

2 A concise Viterbi-like algorithm in CHR

The fundamental parts of the Viterbi algorithm can be expressed in CHR as shown
in fig. 1; the specific HMM is encoded as a set of constraints of the forms
trans(q1,q2,p1) and emit(q3,`,p2), where p1 is the probability to transit from
state q1 to q2, and p2 is the probability to emit the letter ` in state q3. For simplicity
and wlog, we assume a unique initial state, consistently called q0, and that any state
serves as a final state. The intuitive meaning of a constraint path(E,q,p,π) is that
there exists a partial path starting in q0 and ending in q with probability p, and E is
the remaining part of the sequence that needs to be analyzed in order to complete a
full path; for ease of programming, the argument π represents this partial path in re-
versed order. The initial query should be stated as “:- HMM, path(L,q0,1,[])”

:- chr_constraint path/4, trans/3, emit/3.

expand @ trans(Q,Q1,PT), emit(Q,L,PE), path([L|Ls],Q,P,PathRev) ==>

P1 is P*PT*PE, path(Ls,Q1,P1,[Q1|PathRev]).

prune @ path(Ls,Q,P1,_) \ path(Ls,Q,P2,_) <=> P1 >= P2 | true.

Fig. 1. A naive Viterbi-like algorithm in CHR

where HMM is an encoding of the particular HMM and L a sequence to be analyzed.
Termination follows from the fact that the expand rule always reduces the length

of the first argument in the involved path constraint. A correctness proof, which
is left out due to space limitations, can be made by induction showing that prune

will eventually remove any non-optimal path, but always leaves an optimal one for
any prefix of the sequence, and that expand produces all possible extensions of an
optimal path (for any proper prefix). This proof does not need any assumptions
about the order in which the rules are applied.1

Let us informally analyze the time complexity of this program. For simplicity
we count only the number of constraints that are created during the derivation;
for a detailed analysis, we may refer to the methods of [11,6].2 Assuming a naive,
nondeterministic semantics, we may observe derivations that are exponential in the
length of the sequence to be analyzed; this is the case when, e.g., expand is applied
as long as possible, before any application of prune. Our benchmarks (see appendix)
confirm the exponential behaviour; interestingly, when swapping the order of the

1 Notice that the program of fig. 1 is not confluent, although intuitively very close; we
consider this in more detail in section 4 below.

2 Our simplified time complexity measure abstracts away the cost of search and matching
performed by the CHR system.

rules (i.e., prune first), our tests seem to indicate3 a time complexity of O(n4),
although we cannot present a proof for this hypothesis. This is of course far too
slow for any interesting application, and also unsatisfactory as it is known that the
algorithm can run in linear time when written in an imperative language.

3 Fine-grained control by trigger constraints

Linear time complexity requires an optimal interleaving of the expand and prune

rules, so that any path constraint, which will be pruned sooner or later, is not
expanded. We can sketch a class of derivations of linear size by the pseudo-code
shown in fig. 2. As an attempt to obtain a similar flow of control in CHR, and we

seq:= L;
while seq 6= [] do

1) apply expand as long as possible to constraints of form
path(seq,q,p,π), for any q, p and π;

2) apply prune as long as possible;
3) seq:= tail(seq);

Fig. 2. Pseudo-code for optimal control.

introduce what we call trigger constraints by means of which we can control the
detailed procedural semantics of the underlying implementation. Fig. 3 shows an
adaptation of the previous version with trigger constraints. The initial query should
be stated as “:- HMM, path(L,q0,1,[]),trigger(L)”. During execution, the

:- chr_constraint path/4, trans/3, emit/3, trigger/1.

expand @ trans(Q,Q1,PT), emit(Q,L,PE),

path([L|Ls],Q,P,PathRev), trigger([L|Ls]) ==>

P1 is P*PT*PE, path(Ls,Q1,P1,[Q1|PathRev]).

prune @ path(Ls,Q,P1,_) \ path(Ls,Q,P2,_) <=> P1 >= P1 | true.

step @ trigger([_|Ls]) <=> trigger(Ls).

Fig. 3. Viterbi with trigger constraints, version 1.

trigger constraint will refer to decreasing remainders of the sequence, and for each
such iteration provide the relevant applications of expand and prune. This preserves
the logical meaning of the original program, since 1) the trigger constraints are
added only in the head of the original rules, 2) new rules concerning triggers only,
e.g., the step rule, do not unify any arguments, and 3) no derivation is stopped in
a state where the original program would be able to extend the derivation. Notice
that such a proof would need to refer to the operational semantics of the underlying
implementation as well as to the order of the constraints in the initial query.

We sketch an analysis of the time complexity based on the operational semantics
of standard CHR implementations. No rule will execute before trigger(L) is called
in the initial query, and when this happens, expand will apply as long as possible for
any path constraint referring to L similarly to what is expressed in line 1 in fig. 2,
however, interleaved with prune (line 2). When this phase is done, the constraint
trigger(L) reaches the step rule and mutates into trigger(tail(L)) and the pro-
cess repeats for tail(L), and so on, a thus leading to derivations of linear length, as
the number of steps in each such iteration is independent of the sequence length.

3 This and other estimates for time complexity are made by inspecting higher order dif-
ferences for the measured runtimes.

For a fully connected HMM with k states, k2 new path constraints are created in
each iteration, so the length of the entire derivation becomes O(n · k2). However,
the actual time complexity may become higher as we did not count the time for
matching of list arguments in the expand and prune rules, which may, in the worst
case, add another factor n to the time complexity, thus O(n2 · k2). In fact, our run-
time tests shown in the appendix suggest O(n3) for both for a randomly generated
sequence and a worst case sequence that repeats a single letter. The latter implies
that the comparison of two list arguments always traverses the shortest sequence to
the very end; the benchmarks indicate a huge constants factor between the two.

In order to reduce time for matching, we may add a new argument representing
the length of the sequence to path constraints and let the trigger depend on this
length only; the resulting program is shown in fig. 4. Assuming an implementation

:- chr_constraint path/5, trans/3, emit/3, trigger/1.

expand @ trans(Q,Q1,PT), emit(Q,L,PE),

path(N,[L|Ls],Q,P,PathRev), trigger(N) ==>

P1 is P*PT*PE, N1 is N-1, path(N1,Ls,Q1,P1,[Q1|PathRev]).

prune @ path(N,_,Q,P1,_) \ path(N,_,Q,P2,_) <=> P1 >= P2 | true.

step @ trigger(N) <=> N > 0 | N1 is N-1, trigger(N1).

Fig. 4. Viterbi with trigger constraints, version 2, with length arguments.

that applies a suitable indexing on the first argument, we would expect this to lead
to a linear algorithm in the length of the sequence. However, benchmarks indicate
worst case and average complexity of O(n2), which we may hypothesize relates to
a non-optimal search for path constraints. Swapping the expand and prune rules
only changed the figures with a few percent.

To finally overcome these problems and to reach the theoretically best time
complexity for Viterbi in CHR, we needed to add explicit passive declarations4 and
additional code to remove non-current path constraints; our experiments showed
that both additions were necessary. Such a program is shown in fig. 5. We expect that

:- chr_constraint path/5, trans/3, emit/3, trigger/1, zap/1.

expand @ trans(Q,Q1,PT) # Id1, emit(Q,L,PE) # Id2,

path(N,[L|Ls],Q,P,PathRev) # Id3, trigger(N) ==>

P1 is P*PT*PE, N1 is N-1, path(N1,Ls,Q1,P1,[Q1|PathRev])

pragma passive(Id1), passive(Id2), passive(Id3).

prune @ path(N,_,Q,P1,_) \ path(N,_,Q,P2,_) <=> P1 >= P2 | true.

step @ trigger(N) <=> N > 0 | zap(N), N1 is N-1, trigger(N1).

zap(N) \ path(N,_,_,_,_) # Id <=> true pragma passive(Id).

zap(_) <=> true.

Fig. 5. A linear time Viterbi algorithm in CHR; passive declarations and removal of non-
current path constraints.

a detailed analysis can prove linear complexity. Indeed, our benchmarks indicate
that it does stay linear until sequence lengths of more than 10,000 and increases
significantly from around 20,000 and upwards. While a sequence of length 10,000
can be analyzed in 30 sec., it takes 45 minutes for length 100,000. We expect that
this is related to the memory being exhausted due to extreme stack sizes.

4 Passive declarations are a low-level device that suppresses certain firings of rules; for
details, see, e.g., a manual for any of the major CHR versions in Prolog.

4 Conclusion: Methodological considerations, future and
related work

We have shown an implementation of the Viterbi algorithm in CHR, starting from an
abtract and concise specification expansion and pruning. Systematic extensions by
triggers and other techniques lead to an implementation with ideal time complexity.
The Viterbi algorithm represents a larger class of dynamic programming algorithms
for which we believe that our techniques can be applied.

The naive program of fig. 1 is not confluent due to the fact that, when two
paths exist for the same sub-sequence and with the same probability, prune may
nondeterministically get rid of an arbitrary one of them, leading to different new
states that are not logically equivalent. However, it satisfies a requirement that
we may call relative confluence based on an application specific state equivalence
relation. If, for example, two states differs only by the exchange of path(L,p,q,π)
for path(L,p,q,π′), we consider these states equivalent.

Our plans for future work include the formalization of relative confluence and to
generalize known results for confluent programs [10,1] accordingly. We believe that
this can be very useful as many interesting non-confluent programs are relatively
confluent. For confluent programs, it is possible to show as a general result, that
the addition of trigger constraints – satisfying the requirements noticed above for
the program of fig. 3 – preserves the logical meaning of the program as concerns its
original constraints.

Additional optimizations were needed in order to obtain the best time complex-
ity, based on detailed knowledge about the underlying machinery. Our experience
in doing the exercise for the Viterbi algorithm may inspire to more advanced, au-
tomatic analyses and transformations being applied in CHR implementations in
order to promote declarative programming with competitive execution times. We
may also consider the ideal of a true separation of logic and control, so that we
might do with the naive program of fig. 1, complemented by an additional control
specification, which may resemble our abstract algorithm in fig. 2.

An attempt to obtain such a separation have been done by rule priorities [14].
The priority of each rule is expressed in terms of an arithmetic expression referring
to variables in the head of the rules. While the control mechanism appears as dec-
orations to the rules, rather than infiltrating the code as our triggers do, it is also
clear that the rules need to be designed in the first place so that the rule heads ac-
tually contain the necessary information. For the Viterbi algorithm it seems obvious
that the sequence length needs to be present in order to express relevant priorities.
We have not tried to express the Viterbi algorithm in CHR with rule priorities, but
it seems to require advanced algebraic skills to encode the desired control pattern.

There has been other work studying CHR for expressing algorithms [13]. We
will emphasize [16] that gives a detailed analysis of how Dijkstra’s shortest path
algorithm [7] can be implemented in CHR; specifically, the authors studied the use
of priority queues. Similar techniques have been employed by [3] for probabilistic
abductive logic programming in CHR, by [2] for soft constraints and by [15] to
express imperative control constructs in CHR. We have not seen any earlier, sys-
tematic approach for adding detailed procedural control to confluent (or relatively
confluent) programs in order to get the best out of the pruning rules. The Viterbi
algorithm has been formulated as a constraint problem by [4,5], but not in CHR.

Finally, we notice that CHR is suited for describing the fundamental steps of
interesting algorithms, but it is difficult to consider it as a serious implementation
language at present. While the theoretically best complexity often can be reached in
CHR, there is typically a huge constant factor due to the overhead in the underlying
CHR and Prolog runtime systems. For the Viterbi algorithm, it is thought provoking
that it can be implemented very efficiently by a handful of lines of imperative code.

Appendix: Benchmarks

The different variants of the Viterbi algorithm have been tested for a fixed, fully
connected HMM with 4 states (plus a start state that cannot be re-entered) and
an emission alphabet of 4 letters. We have measured runtimes as functions of the
sequence lengths. In most cases, we test on a randomly generated sequence, consid-
ering it as “typical” or “average”. Tests were made with SICStus Prolog 4.0.4 on
a Macintosh 2.4 GHz Intel Core 2 Duo with 4GB RAM, and runtimes have been
measured using SICStus Prolog’s statistics(runtime, . . .) device that ignores
any time spent on garbage collection and other memory management tasks. Run-
times below 10 seconds were taken as average of 10 runs, whereas higher ones were
measured by a single run. Space complexity was not considered.

Fig. 6 shows runtimes for the naive algorithm (fig. 1) for the two alternative
ordering of its rules. The top curve to the left, for the expand rule first, confirms

Fig. 6. Naive algorithm with different rule orders; x-axis n, y-axis ms log-scaled.

our expectation of exponential complexity. Swapping the rules so that prune comes
first, reduces the complexity drastically. We have measured for random sequences
(the “typical”) plus the worst case for this algorithm, which are sequences that
repeat a single letter. The right part shows the prune first version for the two sorts
of sequences for n up to 150. An inspection of higher order differences made from
the actual figures indicates that O(n4) is a reasonable hypothesis for both typical
and worst case; the worst case is about 3 times slower than the typical for n = 150.

Fig. 7 shows runtimes for the versions that use trigger constraints. The two top
curves to the left shows typical and worst case for the version where triggers use
lists, cf. fig. 3. In both cases, differences seem to indicate O(n3); for n = 150 there is
a factor 24 between typical and worst case. The measurements when triggers use the
list length, as opposed to the actual list, is shown as the lowest curve in the left part
and continues as the top curve to the right. Differences suggest O(n2); the typical
and worst case used above provides the same runtimes and swapping the expand

and prune rules changes only a few percent. Finally, the right part of fig. 7 shows

Fig. 7. Algorithm with triggers based on sequences and lengths; x-axes n, y-axes ms log-
scaled in left part, linear in the right part.

also runtimes for the fully optimized version of fig. 5 with passive declarations and
removal of non-current path constraints. It stays linear up to above n = 10, 000,
and for larger values, 20,000 and upwards (not shown), the time grows higher, most
likely because memory begins to be exhausted due to stack sizes. For n = 10, 000,
the time is about half a minute and for n = 100, 000, 45 minutes.

